Shader "Unlit/Shdow"
{
    Properties
    {
        _MainTex ("Texture", 2D) = "white" {}
    }
    SubShader
    {
        Tags { "RenderType"="Opaque" }
        LOD 100

        Pass
        {
            // 采样阴影
            // 1.在v2f中添加UNITY_SHADOW_COORDS(idx),unity会自动声明一个叫_ShadowCoord的float4变量，用作阴影的采样坐标.
            // 2.在顶点着色器中添加TRANSFER_SHADOW(o)，用于将上面定义的_ShadowCoord纹理采样坐标变换到相应的屏幕空间纹理坐标，为采样阴影纹理使用.
            // 3.在片断着色器中添加UNITY_LIGHT_ATTENUATION(atten, i, i.worldPos)，其中atten即存储了采样后的阴影.
            Tags { "LightMode"="ForwardBase" }  // 且采样阴影的时候需要修改渲染路径为 ForwarBase (因为拉近拉远时出现阴影消失的问题)

            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            // make fog work
            #pragma multi_compile_fog

            // 采样阴影时需要对应的变体开关 (这些其实是涉及调用的宏，这些宏里面有需要用到当前这些变体的状态)
            //#pragma multi_compile_fwdbase
            #pragma multi_compile DIRECTIONAL
            #pragma multi_compile SHADOWS_SCREEN
            
            #include "UnityCG.cginc"
            #include "AutoLight.cginc"

            struct appdata
            {
                float4 vertex : POSITION;
            };

            struct v2f
            {
                float4 worldPos : TEXCOORD1;
                float4 pos :SV_POSITION;   // 为什么要改名pos呢？  因为TRANSFER_SHADOW 进入到实际取的是 .pos 的顶点
                // 这里取的参数实际是 TEXCOORD2 unity会自动声明一个叫_ShadowCoord的float4变量，用作阴影的采样坐标.
                // 需要引用  AutoLight.cginc
                UNITY_SHADOW_COORDS(2)
            };

            v2f vert(appdata v)
            {
                v2f o = (v2f)0;
                o.pos = UnityObjectToClipPos(v.vertex);
                o.worldPos = mul(unity_ObjectToWorld,v.vertex);
                // 用于将上面定义的_ShadowCoord纹理采样坐标变换到相应的屏幕空间纹理坐标，为采样阴影纹理使用.
                TRANSFER_SHADOW(o)
                return o;
            }

            fixed4 frag (v2f i) : SV_Target
            {
                // 其中atten即存储了采样后的阴影.
                UNITY_LIGHT_ATTENUATION(atten, i, i.worldPos)
                return atten;
            }
            ENDCG
        }
    }

	FallBack "Diffuse"
}
